FC-4 Fader Control Board

PN: 31205

Document Version: 1.0 • 18 DEC 2007

DESCRIPTION

Whether you build props, design special effects, or work in small-scale lighting control, you have probably had the need to intelligently-control the brightness of one or more 120 VAC lamps. The FC-4 was created for this purpose, giving the designer the ability to dim up to four, 300W incandescent lamps (per board). A simple serial interface allows the host controller to handle up to four FC-4 fader boards on a single control channel. And built-in fade and cross-fade functions simplify the commands required by the host controller.

PRODUCT APPLICATIONS

- Props and holiday displays
- Special FX
- Small-scale lighting control

FEATURES & BENEFITS

- Controls up to four, 300W incandescent lamp channels
- LED on each channel indicates channel level
- Serial control for connection to host microcontroller
- User-configured baud rate (2400 or 38.4k) for serial control
- User-configured address allows up to four FC-4 boards on the same serial connection
- DC Current consumption: 80 mA
- Compact size: 2.5” x 3” (63.5 mm x 76.2 mm)

Mounting the FC-4

WARNING: The FC-4 is designed to switch 120 VAC which can be very dangerous, even lethal, if mishandled. The FC-4 should be mounted before connecting source and switched AC wiring.

Mount the FC-4 on a non-conductive surface using ½” (12 mm) stand-offs and appropriate hardware. Mounting the FC-4 above a conductive (metal) surface could lead to the possibility of a hazardous condition.

DANGER: The FC-4 is capable of switching voltages that can be lethal if mishandled. High voltage AC wiring should be installed by a qualified electrician. Before making any high-voltage connections to the FC-4, ensure that all power has been removed.

Connections – AC Power In and Out

Connect a suitable 120 VAC power cord to the **AC IN** terminal, noting the polarity of the connections (refer to the illustration on page 3). The power cord used must be cable of carrying the expected load of all lamp outputs, not to exceed 1200 watts.

Connect each lamp to the output channel terminals: **O1**, **O2**, **O3**, and **O4**. Ensure that the lamp load on each channel is no greater than 300 watts.

DANGER: Do not connect loads in excess of 300 watts to any output channel of the FC-4. Doing so can damage the FC-4's switching components.

NOTE: The FC-4 is designed for dimming 120 VAC incandescent lamps; do not connect fluorescent lamps or inductive devices like AC motors or solenoids.

Connections – Serial Control

For most applications the FC-4 will be connected to a stand-alone host controller like the Prop-1 or Prop-2. To make the connection between the host and the FC-4 use a 3-wire extender cable (#805-00002). When making the connection it is important to note the color-coded polarity of the extender cable (W = white, R = red, B = black) and match it to the serial header on the FC-4, as well as the output header on the Prop-1, Prop-2, or other host. Note that the FC-4 has two serial connections; these are electrically identical and either may be used for serial-in or serial-thru (to other devices).

Serial Baud Rate Selection

The FC-4 supports two baud rates: 2400 baud for the Prop-1, and 38.4k baud for Prop-2 and other controllers. Baud rate is set with a jumper on the **B/R** header.

B/R	Baud Rate
Out	2400 (Prop-1)
In	38.4k (Prop-2; others)

Serial Address Selection

The serial communications protocol allows the host controller to connect to up to four FC-4 boards; to do this, however, each must have a unique address. The FC-4 board address is set with jumpers on the **A0** and **A1** headers.

A1	A0	Address
Out	Out	0 (%00)
Out	In	1 (%01)
In	Out	2 (%10)
In	In	3 (%11)

Baud Rate and Address Jumpers – A Visual Guide

2400 Baud (Prop-1)				**38.4K Baud** (Prop-2, Prop-SX)			
%00	%01	%10	%11	%00	%01	%10	%11
							

FC-4 in Use

The illustration below shows typical FC-4 connections in use, set to 2400 baud (B/R jumper removed) and address %00 (both address jumpers removed); this is a typical setup for the Prop-1 controller.



It is strongly suggested that each lamp output use separate hot and neutral wires; do not buss the neutral wires together as this could generate excessive heat in a common neutral line.

FC-4 Command Protocol

The FC-4 serial connection conforms to the Parallax AppMod bi-directional communications protocol, which operates at TTL (0 to 5 VDC) using "open-true" baud mode to enable daisy-chaining devices. All exchanges are initiated by the host controller. The format of host communications to the FC-4 is as follows:

"!FC4", *address, command { , data }*

… where "!FC4" is the preamble that allows the FC-4 to exist on the same communications line as other devices (e.g., AP-8, DC-16, etc.), *address* is the board address set with jumpers on A0 and A1, *command* is a single character command designator, and *data* is any value that may be required by command. The following section details valid commands ("V", "G", "L", "S", "P", "A", "D", "X", "F", "C") for the FC-4.

"V" – Get FC-4 Version number

Syntax: "!FC4", *address*, "V"
Reply: 3-byte (ASCII) version string

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!FC4", %00, "V")
SERIN  Sio, OT2400, id0, id1, id2
DEBUG  "FC-4 Version ", #@id0, #@id1, #@id2, CR
```

"G" – Get FC-4 Status

Syntax: "!FC4", *address*, "G"
Reply: Four bytes that represent present output channel levels

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!FC4", %00, "G")
SERIN  Sio, OT2400, lvl1, lvl2, lvl3, lvl4
DEBUG  "Levels = ", #lvl1, " ", #lvl2, " ", #lvl3, " ", #lvl4
```

In this example the present FC-4 channel levels, 0 to 255, are transferred into variables *lvl1* to *lvl4*.

"L" – Set Level for Selected Channel

Syntax: "!FC4", *address*, "L", *chan, level*
– *chan* is output channel: 1..4
– *level* is new brightness: 0 (0%) to 255 (100%)
Reply: none

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!FC4", %00, "L", 2, 128)
```

In this example the lamp output channel O2 will be set to 50% brightness.

"S" – Set Levels for All Channels

Syntax: "!FC4", *address*, "S", *level1, level2, level3, level4*
– *level* is the output brightness of the channel, 0 (0%) to 255 (100%)
Reply: none

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!FC4", %00, "S", 0, 64, 128, 255)
```

In this example the lamp outputs, O1 to O4, will be set to 0%, 25%, 50%, and 100% respectively.

"P" – Preset All Channels

Syntax: "!FC4", *address*, "P", *level*
– *level* is the output brightness of all channels, 0 (0%) to 255 (100%)
Reply: none

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!FC4", %00, "P", 25)
```

In this example the lamp outputs, O1 to O4, will be set to 10%. This can be used to keep filaments warm when fast switching to higher levels is desired.

"A" – All Channels to Full Brightness

Syntax: "!FC4", *address*, "A"
Reply: none

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!FC4", %00, "A")
```

In this example the lamp outputs, O1 to O4, will be set to 100% (full bright).

"D" – Digital Control of All Channels

Syntax: "!FC4", *address*, "D", status
– *status* is the channels output state: %0000 to %1111 (0 bit = off; 1 bit = on)
Reply: none

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!FC4", %00, "D", %0110)
```

In this example the lamp outputs O1 and O4 will be set to 0% (off), and outputs O2 and O3 will be set to 100% (full bright).

"X" – All Channels to Off

Syntax: "!FC4", *address*, "X"
Reply: none

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!FC4", %00, "X")
```

In this example the lamp outputs, O1 to O4, will be set to 0% (off).

"F" – Fade Channel

Syntax: "!FC4", *address*, "F", *chan*, *start, end, stepTm*
– *chan* is output channel: 1..4
– *start* is the initial level, 0 (0%) to 255 (100%)
– *end* is the final level, 0 (0%) to 255 (100%)
– *stepTm* is step timing in milliseconds, 1 to 255
Reply: none

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!FC4", %00, "F", 1, 0, 255, 20)
```

In this example the FC-4 will perform a fade up on channel 1, starting at full off and ending at full on. The duration of the fade is about five seconds. The timing for the fade is the difference between the *start* and *end* levels, plus one, multiplied by the step duration (in milliseconds); 255 – 0 + 1 = 256 x 0.02 = 5.12.

"C", 0 – Cross Fade Two Channels, Program 0 (zero)

Syntax: "!FC4", *address*, "C", 0, *chA*, *chB, stepTm*
- *chA* is output: 1..4; this channel will start at 100% and end at 0%
- *chB* is output: 1..4; this channel will start at 0% and end at 100% (must differ from *chA*)
- *stepTm* is step timing in milliseconds, 1 to 255 (fade takes 256 steps)

Reply: none

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!FC4", %00, "C", 0, 1, 2, 40)
```

In this example the FC-4 will perform a cross-fade between channel 1 (starts at 100%) and channel 2 (starts at 0%). The fade will take approximately ten seconds (256 x 40 milliseconds).

Note: You must specify different channels for *chA* and *chB*, otherwise the command is ignored.

"C", 1 – Cross Fade Two Channels, Program 1

Syntax: "!FC4", *address*, "C", 1, *chA, lvlA, chB, lvlB, stepTm*
- *chA* is output: 1..4; this channel will start at *lvlA* and end at *lvlB*
- *lvlA* is the initial output level for *chA* (0 to 255)
- *chB* is output: 1..4; this channel will start at *lvlB* and end at *lvlA*
- *lvlB* is the initial output level for *chB* (0 to 255)
- *stepTm* is step timing in milliseconds, 1 to 255

Reply: none

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!FC4", %00, "C", 1, 1, 64, 2, 191, 8)
```

In this example the FC-4 will perform a partial cross-fade between channel 1 (starts at 25%) and channel 2 (starts at 75%). The fade will take approximately one second (127 steps x 8 milliseconds).

Cross-fade program one allows the designer to specify the starting levels at other than 0% and 100% as is used with program zero. The timing for the cross-fade will be determined by the difference in levels and the timing per step. To determine the step timing value, divide the number of steps into the desired number of milliseconds for the fade; use the integer result of this division for the step timing.

"C", 2 – Cross Fade Two Channels, Program 2

Syntax: "!FC4", *address*, "C", 2, *chA, lvlA, chB, lvlB, steps, stepTm*
- *chA* is output: 1..4; this channel will start at *lvlA* and end at *lvlA + steps (max level is 255)*
- *lvlA* is the initial output level for *chA* (0 to 255)
- *chB* is output: 1..4; this channel will start at *lvlB* and end at *lvlB – steps (min level is 0)*
- *lvlB* is the initial output level for *chB* (0 to 255)
- *steps* is number of steps in the sequence, 1 to 255
- *stepTm* is step timing in milliseconds, 1 to 255

Reply: none

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!FC4", %00, "C", 2, 1, 0, 2, 255, 128, 8)
```

In this example the FC-4 will perform a partial cross-fade between channel 1 (starts at 0%) and channel 2 (starts at 100%). The fade will take approximately one second (128 steps x 8 milliseconds). Due to their starting values,

channels 1 and 2 will be set to 50% at the end of the cycle. Note that the value of *chA* can never exceed 255 and the value of *chB* can never drop below 0.

Cross-fade program two allows the designer to specify the starting levels at other than 0% and 100% as is used with program zero as well as control the number of steps in the sequence. The overall timing for this partial cross-fade will be determined by the number of steps specified and the timing per step. To determine the step timing value, divide the number of steps into the desired number of milliseconds for the cross-fade; use the integer result of this division for the step timing.

MECHANICAL SPECIFICATIONS



ELECTRICAL SPECIFICATIONS

Parameter	Symbol	Specification			Unit
		Min.	Typical	Max.	
Digital Supply Voltage	V_{IN}	9	12	24	VDC
Digital Supply Current	I_{IN}	90		140	mA
Lamp Supply Voltage	V_{LAMP}	90	120	130	VAC
Lamp Output (per channel)	P_{OUT}	25		300	W
Lamp Output (total)	P_{TOTAL}			1200	W
Operating Temperature	T_{OP}	-40	25	85	C

Specifications subject to change without notice.

Example Program

The following program demonstrates many of the FC-4 features under serial control using the Prop-1 controller. When using the Prop-1 the serial baud rate is set to 2400 (Baud jumper removed).

```
' ================================================================================
'
'   File....... FC-4_Test.BS1
'   Purpose.... FC-4 Features Test
'   Author..... EFX-TEK
'   E-mail..... teamefx@efx-tek.com
'   Started....
'   Updated.... 20 JUL 2007
'
'   {$STAMP BS1}
'   {$PBASIC 1.0}
'
' ================================================================================


' -----[ Program Description ]---------------------------------------------
'
' Demonstration program for the FC-4 Fader Control Board.  All commands are
' sent to the FC-4 through a serial link at 2400 baud (OT2400 baudmode).
' The use of the open baudmode allows boards to be daisy-chained for up to
' 16 fader outputs from one I/O pin.
'
' Command syntax: "!FC4", address, cmd {, data ... }
' -- where "address" is %00 (0) to %11 (3)
' -- "cmd" is a single character
' -- optional "data" is one or more bytes
'
' Note: The ULN2803A interferes with serial transmission to FC-4; remove
'       and replace with ULN2003A (7 channels), leaving P7 contacts open.


' -----[ I/O Definitions ]-------------------------------------------------

SYMBOL  Sio             = 7                     ' SETUP = out, no ULN


' -----[ Variables ]-------------------------------------------------------

SYMBOL  id0             = B0                    ' version string
SYMBOL  id1             = B1
SYMBOL  id2             = B2
SYMBOL  lamp1           = B3                    ' lamp levels
SYMBOL  lamp2           = B4
SYMBOL  lamp3           = B5
SYMBOL  lamp4           = B6
SYMBOL  idx             = B7


' -----[ Initialization ]--------------------------------------------------

Reset:
  DEBUG CLS
  SEROUT Sio, OT2400, ("!!!!!!!FC4", 0, "X")    ' reset FC-4
```

```
' -----[ Program Code ]----------------------------------------------------

Main:
  SEROUT Sio, OT2400, ("!FC4", %00, "V")        ' get version
  SERIN  Sio, OT2400, id0, id1, id2

  DEBUG "FC-4 Version ", #@id0, #@id1, #@id2, CR
  PAUSE 500

All_On:
  DEBUG CR, "All on", CR
  SEROUT Sio, OT2400, ("!FC4", %00, "A")
  GOSUB Show_Status

Dim_All_Manually:
  DEBUG CR,"Fade all (off) with code", CR
  FOR idx = 255 TO 0 STEP -1
    SEROUT Sio, OT2400, ("!FC4", %00, "P", idx)
  NEXT
  GOSUB Show_Status

Digital_Control:
  DEBUG CR, "Digital Control", CR
  FOR idx = %0000 TO %1111
    SEROUT Sio, OT2400, ("!FC4", %00, "D", idx)
    PAUSE 50
    GOSUB Show_Status
  NEXT
  SEROUT Sio, OT2400, ("!FC4", %00, "X")

Fade_Up_One_Auto:
  DEBUG CR, "Fade one channel", CR
  SEROUT Sio, OT2400, ("!FC4", %00, "F", 1, 0, 255, 8)
  PAUSE 2100
  GOSUB Show_Status

Simple_Cross_Fade:
  DEBUG CR, "Simple cross-fade, 1 to 4", CR
  SEROUT Sio, OT2400, ("!FC4", %00, "C", 0, 1, 4, 8)
  PAUSE 2100
  GOSUB Show_Status

  PAUSE 1000
  SEROUT Sio, OT2400, ("!FC4", %00, "X")

The_End:
  DEBUG CR, "FC-4 demo complete."
  END


' -----[ Subroutines ]------------------------------------------------------

Show_Status:
  ' request lamp levels
  SEROUT Sio, OT2400, ("!FC4", %00, "G")
  SERIN  Sio, OT2400, lamp1, lamp2, lamp3, lamp4
  DEBUG  "Levels = ", #lamp1, " ", #lamp2, " ", #lamp3, " ", #lamp4, CR
  RETURN
```

Accessories

750-00007	12 VDC, 1 Amp power supply; *provides ample power for the FC-4*
805-00002	12-inch, 3-pin extension cable; *used to connect external devices to the TTL I/O pins*
27112	Serial Inverter; *used to control the FC-4 and other serial devices from a PC*

Additional Applications

For additional ideas and application notes for the FC-4, be sure to visit us on the Internet at the following links:

www.efx-tek.com
forums.efx-tek.com

For PC-based lighting control, EFX-TEK highly recommends Vixen

www.vixenlights.com

A Vixen driver for the FC-4 is available on our web site.

NOTE: NOTE: The FC-4 makes the dimming of AC lamps very easy, still, caution must be used when installing and using the FC-4 for its intended purpose. EFX-TEK recommends that the FC-4 and related wiring be installed in a suitable enclosure that can be secured to prevent accidental access to live AC circuits.

Consult a qualified electrician regarding the use of AC in your props and displays. For questions on FC-4 application and control, feel free to consult Team EFX at:

teamefx@efx-tek.com

Errata